

SUSPEND IDENTICAL SUPPRESS

SUSPEND IDENTICAL [SUPPRESS] [(rep)]

Dieses Kapitel behandelt folgende Themen:

- Funktion
- Syntax-Beschreibung
- Beispiele

Eine Erläuterung der in dem Syntax-Diagramm verwendeten Symbole entnehmen Sie dem Abschnitt *Syntax-Symbole*.

Verwandte Statements: AT END OF PAGE | AT TOP OF PAGE | CLOSE PRINTER | DEFINE PRINTER | DISPLAY | EJECT | FORMAT | NEWPAGE | PRINT | SKIP | WRITE | WRITE TITLE | WRITE TRAILER

Gehört zur Funktionsgruppe: *Erstellen von Ausgabe-Reports*

Funktion

Mit dem Statement `SUSPEND IDENTICAL SUPPRESS` können Sie den Session-Parameter `IS=ON` (Unterdrückung identischer Feldwerte bei der Ausgabe) für einzelne Datensätze außer Kraft setzen.

Vgl. Session-Parameter `IS` in der *Parameter-Referenz*.

Syntax-Beschreibung

<i>(rep)</i>	<p>Report-Spezifikation:</p> <p>Mit der Notation <i>(rep)</i> kann ein bestimmter anderer Report angegeben werden, auf den sich das <code>SUSPEND IDENTICAL SUPPRESS</code>-Statement beziehen soll.</p> <p>Es kann ein Wert von 0 bis 31 oder ein logischer Name, der mit einem <code>DEFINE PRINTER</code>-Statement zugewiesen wurde, angegeben werden.</p> <p>Falls nichts anderes angegeben wird, bezieht sich das Statement <code>SUSPEND IDENTICAL SUPPRESS</code> auf den ersten Report (Report 0). Informationen darüber, wie Sie das Format eines mit Natural erstellten Ausgabe-Reports steuern, finden Sie im Abschnitt <i>Steuerung der Ausgabe von Daten</i> im <i>Leitfaden zur Programmierung</i>.</p>
--------------	-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

Beispiele

- Beispiel 1 — Programm mit `SUSPEND IDENTICAL SUPPRESS`

- Beispiel 2 — Programm ohne SUSPEND IDENTICAL SUPPRESS

Beispiel 1 — Programm mit SUSPEND IDENTICAL SUPPRESS

```

** Example 'SISEX1': SUSPEND IDENTICAL SUPPRESS
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
*
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
/*
  SUSPEND IDENTICAL SUPPRESS
/*
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    MOVE '***NO CAR***' TO MAKE
  END-NOREC
  DISPLAY NOTITLE
    NAME (RD.) (IS=ON)
    FIRST-NAME (RD.) (IS=ON)
    MAKE (FD.)

  END-FIND
/*
END-READ
END

```

Ausgabe des Programms SISEX1:

NAME	FIRST-NAME	MAKE
JONES	VIRGINIA	CHRYSLER
JONES	MARSHA	CHRYSLER
JONES	ROBERT	CHRYSLER
JONES	LILLY	GENERAL MOTORS
JONES	EDWARD	FORD
JONES	MARTHA	MG
JONES	LAUREL	GENERAL MOTORS
JONES	KEVIN	GENERAL MOTORS
JONES	GREGORY	DATSUN
JONES	EDWARD	FORD
JONES	EDWARD	***NO CAR***
JOPER	MANFRED	***NO CAR***
JOUSSELIN	DANIEL	RENAULT
JUBE	GABRIEL	***NO CAR***
JUNG	ERNST	***NO CAR***
JUNKIN	JEREMY	***NO CAR***

Beispiel 2 — Programm ohne SUSPEND IDENTICAL SUPPRESS

```

** Example 'SISEX2': SUSPEND IDENTICAL SUPPRESS (compare with SISEX1)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
/*
/* SUSPEND IDENTICAL SUPPRESS      /* statement removed
/*
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    MOVE '***NO CAR***' TO MAKE
  END-NOREC
  DISPLAY NOTITLE
    NAME (RD.) (IS=ON)
    FIRST-NAME (RD.) (IS=ON)
    MAKE (FD.)
  END-FIND
/*
END-READ
END

```

Ausgabe des Programms SISEX2:

NAME	FIRST-NAME	MAKE
JONES	VIRGINIA	CHRYSLER
	MARSHA	CHRYSLER
		CHRYSLER
	ROBERT	GENERAL MOTORS
	LILLY	FORD
		MG
	EDWARD	GENERAL MOTORS
	MARTHA	GENERAL MOTORS
	LAUREL	GENERAL MOTORS
	KEVIN	DATSUN
JOPER	GREGORY	FORD
	EDWARD	***NO CAR***
JOUSSELIN	MANFRED	***NO CAR***
	DANIEL	RENAULT
JUBE	GABRIEL	***NO CAR***
JUNG	ERNST	***NO CAR***
JUNKIN	JEREMY	***NO CAR***